

nama : achmad dzul fauzan bahri(24030130061)

Pengantar: Kalkulus Menggunakan EMT

Kalkulus adalah cabang matematika yang mempelajari perubahan dan gerak, meliputi konsep limit, turunan, dan integral. Konsep-konsep ini penting untuk memahami berbagai fenomena dalam ilmu alam, teknik, ekonomi, dan teknologi. Euler Math Toolbox (EMT) digunakan untuk membantu mempelajari Kalkulus secara komputasional dan visual. Dengan EMT, kita dapat mendefinisikan fungsi, menghitung limit, turunan, dan integral secara otomatis, serta menampilkan grafik 2D dan 3D untuk memahami konsep secara lebih intuitif. Pembelajaran Kalkulus dengan EMT membuat proses belajar menjadi lebih interaktif, efisien, dan mudah dipahami karena mahasiswa dapat melihat hubungan antara rumus, grafik, dan hasil perhitungan secara langsung.

Mendefinisikan Fungsi

Terdapat beberapa cara mendefinisikan fungsi pada EMT, yakni:

- Menggunakan format nama_fungsi := rumus fungsi (untuk fungsi numerik),
- Menggunakan format nama_fungsi &= rumus fungsi (untuk fungsi simbolik, namun dapat dihitung secara numerik),
- Menggunakan format nama_fungsi &&= rumus fungsi (untuk fungsi simbolik murni, tidak dapat dihitung langsung),
- Fungsi sebagai program EMT.

Setiap format harus diawali dengan perintah function (bukan sebagai ekspresi).

Berikut adalah beberapa contoh cara mendefinisikan fungsi:

$$f(x) = 2x^2 + e^{\sin(x)}.$$

```
>function f(x) := 2*x^2+exp(sin(x)) // fungsi numerik
>f(0), f(1), f(pi)
```

```
1
4.31977682472
20.7392088022
```

```
>f(a) // tidak dapat dihitung nilainya
```

```
Variable or function a not found.
Error in:
f(a) // tidak dapat dihitung nilainya ...
^
```

Silakan Anda plot kurva fungsi di atas! Berikutnya kita definisikan fungsi:

$$g(x) = \frac{\sqrt{x^2 - 3x}}{x + 1}$$

```
>function g(x) := sqrt(x^2-3*x)/(x+1)
>g(3)
```

0

```
>g(0)
```

0

```
>g(1) // kompleks, tidak dapat dihitung oleh fungsi numerik
```

```
Floating point error!
Error in sqrt
Try "trace errors" to inspect local variables after errors.
g:
  useglobal; return sqrt(x^2-3*x)/(x+1)
Error in:
g(1) // kompleks, tidak dapat dihitung oleh fungsi numerik ...
^
```

Silakan Anda plot kurva fungsi di atas!

```
>f(g(5)) // komposisi fungsi
```

2.20920171961

```
>g(f(5))
```

0.950898070639

```
>function h(x) := f(g(x)) // definisi komposisi fungsi  
>h(5) // sama dengan f(g(5))
```

2.20920171961

Silakan Anda plot kurva fungsi komposisi fungsi f dan g:

$$h(x) = f(g(x))$$

dan

$$u(x) = g(f(x))$$

bersama-sama kurva fungsi f dan g dalam satu bidang koordinat.

```
>f(0:10) // nilai-nilai f(0), f(1), f(2), ..., f(10)
```

```
[1, 4.31978, 10.4826, 19.1516, 32.4692, 50.3833, 72.7562,  
99.929, 130.69, 163.51, 200.58]
```

```
>fmap(0:10) // sama dengan f(0:10), berlaku untuk semua fungsi
```

```
[1, 4.31978, 10.4826, 19.1516, 32.4692, 50.3833, 72.7562,  
99.929, 130.69, 163.51, 200.58]
```

```
>gmap(200:210)
```

```
[0.987534, 0.987596, 0.987657, 0.987718, 0.987778, 0.987837,  
0.987896, 0.987954, 0.988012, 0.988069, 0.988126]
```

Misalkan kita akan mendefinisikan fungsi

$$f(x) = \begin{cases} x^3 & x > 0 \\ x^2 & x \leq 0. \end{cases}$$

Fungsi tersebut tidak dapat didefinisikan sebagai fungsi numerik secara "inline" menggunakan format `:=`, melainkan didefinisikan sebagai program. Perhatikan, kata "map" digunakan agar fungsi dapat menerima vektor sebagai input, dan hasilnya berupa vektor. Jika tanpa kata "map" fungsinya hanya dapat menerima input satu nilai.

```
>function map f(x) ...
```

```
    if x>0 then return x^3
    else return x^2
    endif;
endfunction
```

```
>f(1)
```

```
1
```

```
>f(-2)
```

```
4
```

```
>f(-5:5)
```

```
[25, 16, 9, 4, 1, 0, 1, 8, 27, 64, 125]
```

```
>aspect(1.5); plot2d("f(x)",-5,5):
```

Fungsi simbolik

```
>function f(x) &= 2*E^x // fungsi simbolik
```

$2 E^x$

```
>$f(a) // nilai fungsi secara simbolik  
>f(E) // nilai fungsi berupa bilangan desimal
```

30.308524483

```
>$f(E), $float(%)  
>function g(x) &= 3*x+1
```

$3 x + 1$

```
>function h(x) &= f(g(x)) // komposisi fungsi
```

$$h(x) = \frac{3x + 1}{2}$$

```
>plot2d("h(x)",-1,1):
```

LATIHAN

Bukalah buku Kalkulus. Cari dan pilih beberapa (paling sedikit 5 fungsi berbeda tipe/bentuk/jenis) fungsi dari buku tersebut, kemudian definisikan fungsi-fungsi tersebut dan komposisinya di EMT pada baris-baris perintah berikut (jika perlu tambahkan lagi). Untuk setiap fungsi, hitung beberapa nilainya, baik untuk satu nilai maupun vektor. Gambar grafik fungsi-fungsi tersebut dan komposisi-komposisi 2 fungsi.

Juga, carilah fungsi beberapa (dua) variabel. Lakukan hal sama seperti di atas.

Contoh Penyelesaian:

1. Fungsi linear

$$f(x) = 2x + 3$$

```
>function f1(x) := 2*x+3  
>f1(5)
```

13

```
>f1(0:5)
```

[3, 5, 7, 9, 11, 13]

```
>plot2d("f1(x)",-5,5):
```

2. Fungsi kuadrat

$$f(x) = x^2 - 4x + 3$$

```
>function f2(x) := x^2 - 4*x + 3  
>f2(2), f2(5)
```

```
-1  
8
```

```
>plot2d("f2(x)",-2,6):
```

3. Fungsi trigonometri

$$f(x) = \sin(x) + \cos(x)$$

```
>function f3(x) := sin(x)+cos(x)  
>f3(pi/4)
```

```
1.41421356237
```

```
>plot2d("f3(x)",-2*pi,2*pi):
```

4. Fungsi eksponensial

$$f(x) = e^x$$

```
>function f4(x) := exp(x)
>f4(1), f4(2)
```

```
2.71828182846
7.38905609893
```

```
>plot2d("f4(x)",-2,2):
```

5. Fungsi logaritma

$$f(x) = \ln(x)$$

```
>function f5(x) := ln(x)
>f5(1), f5(E)
```

```
0
1
```

```
>plot2d("f5(x)",0.1,5):
```

Komposisi fungsi (contoh)

$$h(x) = f_2(f_1(x))$$

```
>function h(x) := f2(f1(x))  
>h(2)
```

24

```
>plot2d("h(x)",-5,5):
```

Fungsi dua variabel

$$f(x,y) = x^2 + y^2$$

```
>function fxy(x,y) := x^2 + y^2  
>fxy(1,2)
```

5

```
>plot3d("x^2+y^2",-5,5,-5,5):
```

$$g(x, y) = e^{-(x^2+y^2)}$$

```
>function gxy(x,y) := exp(-(x^2+y^2))  
>gxy(1,1)
```

0.135335283237

```
>plot3d("exp(-(x^2+y^2))",-2,2,-2,2):
```

Menghitung Limit

Perhitungan limit pada EMT dapat dilakukan dengan menggunakan fungsi Maxima, yakni "limit". Fungsi "limit" dapat digunakan untuk menghitung limit fungsi dalam bentuk ekspresi maupun fungsi yang sudah didefinisikan sebelumnya. Nilai limit dapat dihitung pada sebarang nilai atau pada tak hingga (-inf, minf, dan inf). Limit kiri dan limit kanan juga dapat dihitung, dengan cara memberi opsi "plus" atau "minus". Hasil limit dapat berupa nilai, "und" (tak definisi), "ind" (tak tentu namun terbatas), "infinity" (kompleks tak hingga).

Perhatikan beberapa contoh berikut. Perhatikan cara menampilkan perhitungan secara lengkap, tidak hanya menampilkan hasilnya saja.

```
>$showev('limit(sqrt(x^2-3*x)/(x+1),x,inf))  
>$limit((x^3-13*x^2+51*x-63)/(x^3-4*x^2-3*x+18),x,3)
```

maxima: 'limit((x^3-13*x^2+51*x-63)/(x^3-4*x^2-3*x+18),x,3)=limit((x^3-13*x^2+51*x-63)/(x^3-4*x^2-3*x+18),x,3)

Fungsi tersebut diskontinu di titik x=3. Berikut adalah grafik fungsinya.

```
>aspect(1.5); plot2d("(x^3-13*x^2+51*x-63)/(x^3-4*x^2-3*x+18)",0,4); plot2d(3,-4/5,>points,style="ow  
>$limit(2*x*sin(x)/(1-cos(x)),x,0)
```

maxima: `'limit(2*x*sin(x)/(1-cos(x)),x,0)=limit(2*x*sin(x)/(1-cos(x)),x,0)`

Fungsi tersebut diskontinu di titik $x=0$. Berikut adalah grafiknya.

```
>plot2d("2*x*sin(x)/(1-cos(x))",-pi,pi); plot2d(0,4,>points,style="ow",>add):  
>$limit(cot(7*h)/cot(5*h),h,0)
```

maxima: `showev('limit(cot(7*h)/cot(5*h),h,0))`

Fungsi tersebut juga diskontinu (karena tidak terdefinisi) di $x=0$. Berikut adalah grafiknya.

```
>plot2d("cot(7*x)/cot(5*x)",-0.001,0.001); plot2d(0,5/7,>points,style="ow",>add):  
>$showev('limit(((x/8)^(1/3)-1)/(x-8),x,8))
```

Tunjukkan limit tersebut dengan grafik, seperti contoh-contoh sebelumnya.

```
>$showev('limit(1/(2*x-1),x,0))
```

Tunjukkan limit tersebut dengan grafik, seperti contoh-contoh sebelumnya.

```
>$showev('limit((x^2-3*x-10)/(x-5),x,5))
```

Tunjukkan limit tersebut dengan grafik, seperti contoh-contoh sebelumnya.

```
>$showev('limit(sqrt(x^2+x)-x,x,inf))
```

Tunjukkan limit tersebut dengan grafik, seperti contoh-contoh sebelumnya.

```
>$showev('limit(abs(x-1)/(x-1),x,1,minus))
```

Hitung limit di atas untuk x menuju 1 dari kanan.

Tunjukkan limit tersebut dengan grafik, seperti contoh-contoh sebelumnya.

```
>$showev('limit(sin(x)/x,x,0))  
>plot2d("sin(x)/x",-pi,pi); plot2d(0,1,>points,style="ow",>add):  
>$showev('limit(sin(x^3)/x,x,0))
```

Tunjukkan limit tersebut dengan grafik, seperti contoh-contoh sebelumnya.

```

>$showev('limit(log(x), x, minf))
>$showev('limit((-2)^x,x, inf))
>$showev('limit(t-sqrt(2-t),t,2,minus))
>$showev('limit(t-sqrt(2-t),t,2,plus))
>$showev('limit(t-sqrt(2-t),t,5,plus)) // Perhatikan hasilnya
>plot2d("x-sqrt(2-x)",0,2):
>$showev('limit((x^2-9)/(2*x^2-5*x-3),x,3))

```

Tunjukkan limit tersebut dengan grafik, seperti contoh-contoh sebelumnya.

```

>$showev('limit((1-cos(x))/x,x,0))

```

Tunjukkan limit tersebut dengan grafik, seperti contoh-contoh sebelumnya.

```

>$showev('limit((x^2+abs(x))/(x^2-abs(x)),x,0))

```

Tunjukkan limit tersebut dengan grafik, seperti contoh-contoh sebelumnya.

```

>$showev('limit((1+1/x)^x,x,inf))
>plot2d("(1+1/x)^x",0,1000):
>$showev('limit((1+k/x)^x,x,inf))
>$showev('limit((1+x)^(1/x),x,0))

```

Tunjukkan limit tersebut dengan grafik, seperti contoh-contoh sebelumnya.

```
>$showev('limit((x/(x+k))^x,x,inf))  
>$showev('limit((E^x-E^2)/(x-2),x,2))
```

Tunjukkan limit tersebut dengan grafik, seperti contoh-contoh sebelumnya.

```
>$showev('limit(sin(1/x),x,0))  
>$showev('limit(sin(1/x),x,inf))  
>plot2d("sin(1/x)",-0.001,0.001):
```

Berikut adalah contoh-contoh fungsi untuk latihan perhitungan limit di EMT. Setiap fungsi dituliskan rumusnya, dijabarkan pengerjaan limitnya secara manual, lalu dihitung menggunakan EMT, dan terakhir ditampilkan grafik untuk konfirmasi.

1. Fungsi rasional

```
>plot2d("(x^2-4)/(x-2)",0,4);  
>plot2d(2,4,>points,style="ow",>add):
```

2. Fungsi akar

```
>plot2d("(sqrt(x+1)-1)/x",-1,3);  
>plot2d(0,0.5,>points,style="ow",>add):
```

3. Fungsi trigonometri

```
>plot2d("sin(5*x)/(2*x)",-pi,pi);  
>plot2d(0,2.5,>points,style="ow",>add):
```

4. Fungsi eksponensial

```
>plot2d("(1+1/x)^x",1,1000):
```

5. Fungsi logaritma

```
>plot2d("log(x+1)/x",0,1);  
>plot2d(0,1,>points,style="ow",>add):
```

Turunan Fungsi

Definisi turunan:

$$f'(x) = \lim_{h \rightarrow 0} \frac{f(x+h) - f(x)}{h}$$

Berikut adalah contoh-contoh menentukan turunan fungsi dengan menggunakan definisi turunan (limit).

```
>$showev('limit(((x+h)^2-x^2)/h,h,0)) // turunan x^2
>p &= expand((x+h)^2-x^2)|simplify; $p //pembilang dijabarkan dan disederhanakan
>q &=ratsimp(p/h); $q // ekspresi yang akan dihitung limitnya disederhanakan
>$limit(q,h,0) // nilai limit sebagai turunan
>$showev('limit(((x+h)^n-x^n)/h,h,0)) // turunan x^n
```

Mengapa hasilnya seperti itu? Tuliskan atau tunjukkan bahwa hasil limit tersebut benar, sehingga benar turunan fungsinya benar. Tulis penjelasan Anda di komentar ini.

Sebagai petunjuk, ekspansikan $(x+h)^n$ dengan menggunakan teorema binomial.

```
>$showev('limit((sin(x+h)-sin(x))/h,h,0)) // turunan sin(x)
```

Mengapa hasilnya seperti itu? Tuliskan atau tunjukkan bahwa hasil limit tersebut benar, sehingga benar turunan fungsinya benar. Tulis penjelasan Anda di komentar ini.

Sebagai petunjuk, ekspansikan $\sin(x+h)$ dengan menggunakan rumus jumlah dua sudut.

```
>$showev('limit((log(x+h)-log(x))/h,h,0)) // turunan log(x)
```

Mengapa hasilnya seperti itu? Tuliskan atau tunjukkan bahwa hasil limit tersebut benar, sehingga benar turunan fungsinya benar. Tulis penjelasan Anda di komentar ini.

Sebagai petunjuk, gunakan sifat-sifat logaritma dan hasil limit pada bagian sebelumnya di atas.

```
>$showev('limit((1/(x+h)-1/x)/h,h,0)) // turunan 1/x
>$showev('limit((E^(x+h)-E^x)/h,h,0)) // turunan f(x)=e^x
```

```
Answering "Is x an integer?" with "integer"
Answering "Is x an integer?" with "integer"
Answering "Is x an integer?" with "integer"
Answering "Is x an integer?" with "integer"
Answering "Is x an integer?" with "integer"
Maxima is asking
Acceptable answers are: yes, y, Y, no, n, N, unknown, uk
Is x an integer?

Use assume!
Error in:
```

~

Maxima bermasalah dengan limit:

$$\lim_{h \rightarrow 0} \frac{e^{x+h} - e^x}{h}.$$

Oleh karena itu diperlukan trik khusus agar hasilnya benar.

```
>$showev('limit((E^h-1)/h,h,0))  
>$showev('factor(E^(x+h)-E^x))  
>$showev('limit(factor((E^(x+h)-E^x)/h),h,0)) // turunan f(x)=e^x  
>function f(x) &= x^x
```

x
x

```
>$showev('limit(f(x),x,0))
```

Silakan Anda gambar kurva

$$y = x^x.$$

```
>$showev('limit((f(x+h)-f(x))/h,h,0)) // turunan f(x)=x^x
```

Di sini Maxima juga bermasalah terkait limit:

$$\lim_{h \rightarrow 0} \frac{(x+h)^{x+h} - x^x}{h}.$$

Dalam hal ini diperlukan asumsi nilai x .

```
>&assume(x>0); $showev('limit((f(x+h)-f(x))/h,h,0)) // turunan f(x)=x^x
```

Mengapa hasilnya seperti itu? Tuliskan atau tunjukkan bahwa hasil limit tersebut benar, sehingga benar turunan fungsinya benar. Tulis penjelasan Anda di komentar ini.

```
>&forget(x>0) // jangan lupa, lupakan asumsi untuk kembali ke semula
```

$[x > 0]$

```
>&forget(x<0)
```

$[x < 0]$

```
>&facts()
```

```
[]
```

```
>$showev('limit((asin(x+h)-asin(x))/h,h,0)) // turunan arcsin(x)
```

Mengapa hasilnya seperti itu? Tuliskan atau tunjukkan bahwa hasil limit tersebut benar, sehingga benar turunan fungsinya benar. Tulis penjelasan Anda di komentar ini.

```
>$showev('limit((tan(x+h)-tan(x))/h,h,0)) // turunan tan(x)
```

Mengapa hasilnya seperti itu? Tuliskan atau tunjukkan bahwa hasil limit tersebut benar, sehingga benar turunan fungsinya benar. Tulis penjelasan Anda di komentar ini.

```
>function f(x) &= sinh(x) // definisikan f(x)=sinh(x)
```

```
sinh(x)
```

```
>function df(x) &= limit((f(x+h)-f(x))/h,h,0); $df(x) // df(x) = f'(x)
```

Hasilnya adalah $\cosh(x)$, karena

$$\frac{e^x + e^{-x}}{2} = \cosh(x).$$

```
>plot2d(["f(x)","df(x)"],-pi,pi,color=[blue,red]):  
>function f(x) &= sin(3*x^5+7)^2
```

$$\sin^2(3x^5 + 7)$$

```
>diff(f,3), diffc(f,3)
```

```
1198.32948904  
1198.72863721
```

Apakah perbedaan diff dan diffc?

```

>$showev('diff(f(x),x))
>$% with x=3
>$float(%)
>plot2d(f,0,3.1):
>function f(x) &=5*cos(2*x)-2*x*sin(2*x) // mendefinisikan fungsi f

```

$$5 \cos(2 x) - 2 x \sin(2 x)$$

```

>function df(x) &=diff(f(x),x) // fd(x) = f'(x)

```

$$- 12 \sin(2 x) - 4 x \cos(2 x)$$

```

>$'f(1)=f(1), $float(f(1)), $'f(2)=f(2), $float(f(2)) // nilai f(1) dan f(2)
>xp=solve("df(x)",1,2,0) // solusi f'(x)=0 pada interval [1, 2]

```

$$1.35822987384$$

```

>df(xp), f(xp) // cek bahwa f'(xp)=0 dan nilai ekstrim di titik tersebut

```

0

-5.67530133759

```
>plot2d(["f(x)","df(x)"],0,2*pi,color=[blue,red]): //grafik fungsi dan turunannya
```

Pilihlah minimal 5 fungsi dari buku Kalkulus. Untuk setiap fungsi: 1. Definisikan fungsinya di EMT. 2. Hitung turunannya dengan:

- (a) Definisi limit
- (b) Perintah diff
- (c) Perhitungan manual dengan Maxima

3. Gambarkan grafik fungsi asli dan turunannya pada koordinat

yang sama (warna merah = turunan).

1. Fungsi polinomial

```
>function f(x) := x^2+3*x+2
```

(a) Turunan dengan definisi limit

```
>$limit((f(x+h)-f(x))/h,h,0)
```

(b) Turunan dengan perintah diff

```
>$diff(f(x),x)
```

(c) Turunan simbolik dengan Maxima

```
>&diff(x^2+3*x+2,x)
```

$$2x + 3$$

Grafik fungsi dan turunannya

```
>plot2d("x^2+3*x+2", -5, 5)  
>plot2d("2*x+3", -5, 5, >add, color=red)
```

2. Fungsi akar

```
>function g(x) := sqrt(x+1)
```

(a) Turunan dengan definisi limit

```
>$limit((g(x+h)-g(x))/h,h,0)
```

(b) Turunan dengan perintah diff

```
>$diff(g(x),x)
```

(c) Turunan simbolik dengan Maxima

```
>&diff(sqrt(x+1),x)
```

$$\frac{1}{2 \sqrt{x + 1}}$$

Grafik fungsi dan turunannya

```
>plot2d("sqrt(x+1)", -1, 5)  
>plot2d("1/(2*sqrt(x+1))", -1, 5, >add, color=red)
```

3. Fungsi trigonometri

```
>function h(x) := sin(x)
```

(a) Turunan dengan definisi limit

```
>$limit((sin(x+h)-sin(x))/h, h, 0)
```

(b) Turunan dengan perintah diff

```
>$diff(h(x),x)
```

(c) Turunan simbolik dengan Maxima

```
>&diff(sin(x),x)
```

$\cos(x)$

Grafik fungsi dan turunannya

```
>plot2d("sin(x)", -2*pi, 2*pi):  
>plot2d("cos(x)", -2*pi, 2*pi, >add, color=red):
```

4. Fungsi eksponensial

```
>function k(x) := exp(x)
```

(a) Turunan dengan definisi limit

```
>$limit((k(x+h)-k(x))/h,h,0)
```

(b) Turunan dengan perintah diff

```
>$diff(k(x),x)
```

(c) Turunan simbolik dengan Maxima

```
>&diff(exp(x),x)
```

x
E

Grafik fungsi dan turunannya

```
>plot2d("exp(x)", -2, 2):  
>plot2d("exp(x)", -2, 2, >add, color=red):
```

5. Fungsi logaritma

```
>function m(x) := log(x)
```

(a) Turunan dengan definisi limit

```
>$limit((m(x+h)-m(x))/h,h,0)
```

(b) Turunan dengan perintah diff

```
>$diff(m(x),x)
```

(c) Turunan simbolik dengan Maxima

```
>&diff(log(x),x)
```

$$\frac{1}{x}$$

Grafik fungsi dan turunannya

```
>plot2d("log(x)", 0.1, 5):  
>plot2d("1/x", 0.1, 5, >add, color=red):
```

EMT dapat digunakan untuk menghitung integral, baik integral tak tentu maupun integral tentu. Untuk integral tak tentu (simbolik) sudah tentu EMT menggunakan Maxima, sedangkan untuk perhitungan integral tentu EMT sudah menyediakan beberapa fungsi yang mengimplementasikan algoritma kuadratur (perhitungan integral tentu menggunakan metode numerik).

Pada notebook ini akan ditunjukkan perhitungan integral tentu dengan menggunakan Teorema Dasar Kalkulus:

$$\int_a^b f(x) dx = F(b) - F(a), \quad \text{dengan } F'(x) = f(x).$$

Fungsi untuk menentukan integral adalah `integrate`. Fungsi ini dapat digunakan untuk menentukan, baik integral tentu maupun tak tentu (jika fungsinya memiliki antiderivatif). Untuk perhitungan integral tentu fungsi `integrate` menggunakan metode numerik (kecuali fungsinya tidak integrabel, kita tidak akan menggunakan metode ini).

```
>$showev('integrate(x^n,x))
```

Answering "Is n equal to -1?" with "no"

```
>$showev('integrate(1/(1+x),x))
>$showev('integrate(1/(1+x^2),x))
>$showev('integrate(1/sqrt(1-x^2),x))
>$showev('integrate(sin(x),x,0,pi))
>plot2d("sin(x)",0,2*pi):
>$showev('integrate(sin(x),x,a,b))
>$showev('integrate(x^n,x,a,b))
```

Answering "Is n positive, negative or zero?" with "positive"

```
>$showev('integrate(x^2*sqrt(2*x+1),x))
>$showev('integrate(x^2*sqrt(2*x+1),x,0,2))
>$ratsimp(%)
>$showev('integrate((sin(sqrt(x)+a)*E^sqrt(x))/sqrt(x),x,0,pi^2))
>$factor(%)
>function map f(x) &= E^(-x^2)
```

$$E^{-x^2}$$

```
>$showev('integrate(f(x),x))
```

Fungsi f tidak memiliki antiturunan, integralnya masih memuat integral lain.

$$erf(x) = \int \frac{e^{-x^2}}{\sqrt{\pi}} dx.$$

Kita tidak dapat menggunakan teorema Dasar kalkulus untuk menghitung integral tentu fungsi tersebut jika semua batasnya berhingga. Dalam hal ini dapat digunakan metode numerik (rumus kuadratur).

Misalkan kita akan menghitung:

maxima: `'integrate(f(x),x,0,pi)`

```
>x=0:0.1:pi-0.1; plot2d(x,f(x+0.1),>bar); plot2d("f(x)",0,pi,>add):
```

Integral tentu

maxima: `'integrate(f(x),x,0,pi)`

dapat dihampiri dengan jumlah luas persegi-persegi panjang di bawah kurva $y=f(x)$ tersebut. Langkah-langkahnya adalah sebagai berikut.

```
>t &= makelist(a,a,0,pi-0.1,0.1); // t sebagai list untuk menyimpan nilai-nilai x
>fx &= makelist(f(t[i]+0.1),i,1,length(t)); // simpan nilai-nilai f(x)
>// jangan menggunakan x sebagai list, kecuali Anda pakar Maxima!
```

Hasilnya adalah:

maxima: `'integrate(f(x),x,0,pi) = 0.1*sum(fx[i],i,1,length(fx))`

Jumlah tersebut diperoleh dari hasil kali lebar sub-subinterval ($=0.1$) dan jumlah nilai-nilai $f(x)$ untuk $x = 0.1, 0.2, 0.3, \dots, 3.2$.

```
>0.1*sum(f(x+0.1)) // cek langsung dengan perhitungan numerik EMT
```

0.836219610253

Untuk mendapatkan nilai integral tentu yang mendekati nilai sebenarnya, lebar sub-intervalnya dapat diperkecil lagi, sehingga daerah di bawah kurva tertutup semuanya, misalnya dapat digunakan lebar subinterval 0.001. (Silakan dicoba!)

Meskipun Maxima tidak dapat menghitung integral tentu fungsi tersebut untuk batas-batas yang berhingga, namun integral tersebut dapat dihitung secara eksak jika batas-batasnya tak hingga. Ini adalah salah satu keajaiban di dalam matematika, yang terbatas tidak dapat dihitung secara eksak, namun yang tak hingga malah dapat dihitung secara eksak.

```
>$showev('integrate(f(x),x,0,inf))
```

Tunjukkan kebenaran hasil di atas!

Berikut adalah contoh lain fungsi yang tidak memiliki antiderivatif, sehingga integral tentunya hanya dapat dihitung dengan metode numerik.

```
>function f(x) &= x^x
```



```
>$showev('integrate(f(x),x,0,1))  
>x=0:0.1:1-0.01; plot2d(x,f(x+0.01),>bar); plot2d("f(x)",0,1,>add):
```

Maxima gagal menghitung integral tentu tersebut secara langsung menggunakan perintah integrate. Berikut kita lakukan seperti contoh sebelumnya untuk mendapat hasil atau pendekatan nilai integral tentu tersebut.

```
>t &= makelist(a,a,0,1-0.01,0.01);  
>fx &= makelist(f(t[i]+0.01),i,1,length(t));
```

maxima: `'integrate(f(x),x,0,1) = 0.01*sum(fx[i],i,1,length(fx))`

Apakah hasil tersebut cukup baik? perhatikan gambarnya.

```
>function f(x) &= sin(3*x^5+7)^2
```

$$\sin^2(3x^5 + 7)$$

```
>integrate(f,0,1)
```

0.542581176074

```
>&showev('integrate(f(x),x,0,1))
```

$$\int_0^1 \sin^2(3x^5 + 7) dx = \frac{\frac{1}{5} \gamma(-\frac{1}{5}) \sin(14) \sin(\frac{\pi}{10})}{10^{\frac{1}{5}}}$$

$$\begin{aligned}
& - \left(\left(6 \frac{\gamma_{\text{incomplete}}(-, 6 I)}{5} + 6 \frac{\gamma_{\text{incomplete}}(-, -6 I)}{5} \right. \right. \\
& \left. \sin(14) + \left(6 \frac{I \gamma_{\text{incomplete}}(-, 6 I)}{5} \right. \right. \\
& \left. \left. - 6 \frac{I \gamma_{\text{incomplete}}(-, -6 I)}{5} \right) \cos(14) \right) \sin\left(\frac{\pi}{10} - 60\right) / 120
\end{aligned}$$

```
>&float(%)
```

```

      1.0
      /
      [      2      5
      I      sin (3.0 x  + 7.0) dx =
      ]
      /
      0.0
0.09820784258795788 - 0.008333333333333333
(0.3090169943749474 (0.1367372182078336
(4.192962712629476 I gamma__incomplete(0.2, 6.0 I)
- 4.192962712629476 I gamma__incomplete(0.2, - 6.0 I))
+ 0.9906073556948704 (4.192962712629476 gamma__incomplete(0.2, 6.0 I)
+ 4.192962712629476 gamma__incomplete(0.2, - 6.0 I))) - 60.0)

```

```
>$showev('integrate(x*exp(-x),x,0,1)) // Integral tentu (eksak)
```

LATIHAN INTEGRAL

1. Integral tak tentu fungsi polinomial

```
>$showev('integrate(2*x^3+3*x^2+4,x))
```

2. Integral tak tentu fungsi trigonometri

```
>$showev('integrate(sin(x),x))
```

3. Integral tak tentu fungsi rasional

```
>$showev('integrate(1/(1+x),x))
```

4. Integral tentu fungsi eksponensial

```
>$showev('integrate(exp(x),x,0,1))
```

5. Integral tentu fungsi akar

```
>$showev('integrate(sqrt(x),x,0,4)')
```

Aplikasi Integral Tentu

```
>plot2d("x^3-x",-0.1,1.1); plot2d("-x^2",>add); ...  
>b=solve("x^3-x+x^2",0.5); x=linspace(0,b,200); xi=flipx(x); ...  
>plot2d(x|xi,x^3-x|-xi^2,>filled,style="|",fillcolor=1,>add): // Plot daerah antara 2 kurva  
>a=solve("x^3-x+x^2",0), b=solve("x^3-x+x^2",1) // absis titik-titik potong kedua kurva
```

```
0  
0.61803398875
```

```
>integrate("(-x^2)-(x^3-x)",a,b) // luas daerah yang diarsir
```

```
0.0758191713542
```

Hasil tersebut akan kita bandingkan dengan perhitungan secara analitik.

```
>a &= solve((-x^2)-(x^3-x),x); $a // menentukan absis titik potong kedua kurva secara eksak  
>$showev('integrate(-x^2-x^3+x,x,0,(sqrt(5)-1)/2)) // Nilai integral secara eksak  
>$float(%)
```

LATIHAN SOAL

1. Hitung integral tentu dari fungsi $f(x) = x^2$ pada selang $[0,2]$.

```
>integrate("x^2",0,2)
```

2.666666666667

Hasil analitik:

```
>&integrate(x^2,x,0,2)
```

8
-
3

```
>$float(%)  
>plot2d("x^2",0,2):  
>plot2d("0",>add):  
>plot2d([0,2],[0,0],>add):
```

2. Hitung integral tentu dari $f(x) = \sin(x)$ pada selang $[0, \pi]$.

```
>integrate("sin(x)",0,pi)
```

2

Hasil analitik:

```
>&integrate(sin(x),x,0,%pi)
```

2

```
>$float(%)  
>plot2d("sin(x)",0,pi):  
>plot2d("0",>add):
```

3. Hitung integral tentu dari $f(x) = e^x$ pada selang $[0,1]$.

```
>integrate("exp(x)",0,1)
```

1.71828182846

Hasil analitik:

```
>&integrate(exp(x),x,0,1)
```

E - 1

```
>$float(%)  
>plot2d("exp(x)",0,1):  
>plot2d("0",>add):
```

4. Hitung luas daerah antara $f(x) = x^2$ dan $g(x) = x$ pada selang $[0,1]$.

```
>integrate("x-x^2",0,1)
```

0.1666666666667

Hasil analitik:

```
>&integrate(x-x^2,x,0,1)
```

1
-
6

```
>$float(%)  
>plot2d("x^2",0,1):  
>plot2d("x",0,1,>add,color=red):
```

5. Hitung luas daerah antara $f(x) = \cos(x)$ dan sumbu-x pada selang $[0,\pi/2]$.

```
>integrate("cos(x)",0,pi/2)
```

1

Hasil analitik:

```
>&integrate(cos(x),x,0,%pi/2)
```

1

```
>$float(%)  
>plot2d("cos(x)",0,pi/2):  
>plot2d("0",>add):
```

Panjang Kurva

Hitunglah panjang kurva berikut ini dan luas daerah di dalam kurva tersebut.

$$\gamma(t) = (r(t) \cos(t), r(t) \sin(t))$$

dengan

$$r(t) = 1 + \frac{\sin(3t)}{2}, \quad 0 \leq t \leq 2\pi.$$

```
>t=linspace(0,2pi,1000); r=1+sin(3*t)/2; x=r*cos(t); y=r*sin(t); ...  
>plot2d(x,y,>filled,fillcolor=red,style="/",r=1.5): // Kita gambar kurvanya terlebih dahulu  
>function r(t) &= 1+sin(3*t)/2; $'r(t)=r(t)
```

```
>function fx(t) &= r(t)*cos(t); $'fx(t)=fx(t)  
>function fy(t) &= r(t)*sin(t); $'fy(t)=fy(t)  
>function ds(t) &= trigreduce(radcan(sqrt(diff(fx(t),t)^2+diff(fy(t),t)^2))); $'ds(t)=ds(t)  
>$integrate(ds(x),x,0,2*pi) //panjang (keliling) kurva
```

Maxima gagal melakukan perhitungan eksak integral tersebut.

Berikut kita hitung integralnya secara numerik dengan perintah EMT.

```
>integrate("ds(x)",0,2*pi)
```

9.0749467823

Spiral Logaritmik

$$x = e^{ax} \cos x, \quad y = e^{ax} \sin x.$$

```
>a=0.1; plot2d("exp(a*x)*cos(x)","exp(a*x)*sin(x)",r=2,xmin=0,xmax=2*pi):  
>&kill(a) // hapus ekspresi a
```

done

```
>function fx(t) &= exp(a*t)*cos(t); $'fx(t)=fx(t)  
>function fy(t) &= exp(a*t)*sin(t); $'fy(t)=fy(t)  
>function df(t) &= trigreduce(radcan(sqrt(diff(fx(t),t)^2+diff(fy(t),t)^2))); $'df(t)=df(t)  
>S &=integrate(df(t),t,0,2*%pi); $S // panjang kurva (spiral)  
>S(a=0.1) // Panjang kurva untuk a=0.1
```

8.78817491636

Soal:

Tunjukkan bahwa keliling lingkaran dengan jari-jari r adalah $K=2\pi.r$.

Berikut adalah contoh menghitung panjang parabola.

```
>plot2d("x^2",xmin=-1,xmax=1):  
>$showev('integrate(sqrt(1+diff(x^2,x)^2),x,-1,1))  
>$float(%)  
>x=-1:0.2:1; y=x^2; plot2d(x,y); ...  
> plot2d(x,y,points=1,style="o#",add=1):
```

Panjang tersebut dapat dihamperi dengan menggunakan jumlah panjang ruas-ruas garis yang menghubungkan titik-titik pada parabola tersebut.

```
>i=1:cols(x)-1; sum(sqrt((x[i+1]-x[i])^2+(y[i+1]-y[i])^2))
```

2.95191957027

Hasilnya mendekati panjang yang dihitung secara eksak. Untuk mendapatkan hampiran yang cukup akurat, jarak antar titik dapat diperkecil, misalnya 0.1, 0.05, 0.01, dan seterusnya. Cobalah Anda ulangi perhitungannya dengan nilai-nilai tersebut.

Koordinat Kartesius

Berikut diberikan contoh perhitungan panjang kurva menggunakan koordinat Kartesius. Kita akan hitung panjang kurva dengan persamaan implisit:

$$x^3+y^3-3xy=0.$$

```
>z &= x^3+y^3-3*x*y; $z  
>plot2d(z,r=2,level=0,n=100):
```

Kita tertarik pada kurva di kuadran pertama.

```
>plot2d(z,a=0,b=2,c=0,d=2,level=[-10;0],n=100,contourwidth=3,style="/"):
```

Kita selesaikan persamaannya untuk x.

```
>$z with y=1*x, sol &= solve(%,x); $sol
```

Kita gunakan solusi tersebut untuk mendefinisikan fungsi dengan Maxima.

```
>function f(1) &= rhs(sol[1]); $'f(1)=f(1)
```

Fungsi tersebut juga dapat digunakan untuk menggambar kurvanya. Ingat, bahwa fungsi tersebut adalah nilai x dan nilai $y=l*x$, yakni $x=f(l)$ dan $y=l*f(l)$.

```
>plot2d(&f(x),&x*f(x),xmin=-0.5,xmax=2,a=0,b=2,c=0,d=2,r=1.5):
```

Elemen panjang kurva adalah:

$$ds = \sqrt{f'(l)^2 + (lf'(l) + f(l))^2}.$$

```
>function ds(l) &= ratsimp(sqrt(diff(f(l),l)^2+diff(l*f(l),l)^2)); $'ds(l)=ds(l)
>$integrate(ds(l),l,0,1)
```

Integral tersebut tidak dapat dihitung secara eksak menggunakan Maxima. Kita hitung integral tersebut secara numerik dengan Euler. Karena kurva simetris, kita hitung untuk nilai variabel integrasi dari 0 sampai 1, kemudian hasilnya dikalikan 2.

```
>2*integrate("ds(x)",0,1)
```

4.91748872168

```
>2*romberg(&ds(x),0,1)// perintah Euler lain untuk menghitung nilai hampiran integral yang sama
```

4.91748872168

Perhitungan di atas dapat dilakukan untuk sebarang fungsi x dan y dengan mendefinisikan fungsi EMT, misalnya kita beri nama panjangkurva. Fungsi ini selalu memanggil Maxima untuk menurunkan fungsi yang diberikan.

```
>function panjangkurva(fx,fy,a,b) ...
```

```
ds=mxm("sqrt(diff(@fx,x)^2+diff(@fy,x)^2)");  
return romberg(ds,a,b);  
endfunction
```

```
>panjangkurva("x","x^2",-1,1) // cek untuk menghitung panjang kurva parabola sebelumnya
```

2.95788571509

Bandingkan dengan nilai eksak di atas.

```
>2*panjangkurva(mxm("f(x)"),mxm("x*f(x)"),0,1) // cek contoh terakhir, bandingkan hasilnya!
```

4.91748872168

Kita hitung panjang spiral Archimides berikut ini dengan fungsi tersebut.

```
>plot2d("x*cos(x)","x*sin(x)",xmin=0,xmax=2*pi,square=1):  
>panjangkurva("x*cos(x)","x*sin(x)",0,2*pi)
```

21.2562941482

Berikut kita definisikan fungsi yang sama namun dengan Maxima, untuk perhitungan eksak.

```
>&kill(ds,x,fx,fy)
```

done

```
>function ds(fx,fy) &&= sqrt(diff(fx,x)^2+diff(fy,x)^2)
```

$$\sqrt{\text{diff}(fy, x)^2 + \text{diff}(fx, x)^2}$$

```
>sol &= ds(x*cos(x),x*sin(x)); $sol // Kita gunakan untuk menghitung panjang kurva terakhir di atas
>$sol | trigreduce | expand, $integrate(%,x,0,2*pi), %()
```

21.2562941482

Hasilnya sama dengan perhitungan menggunakan fungsi EMT.

Berikut adalah contoh lain penggunaan fungsi Maxima tersebut.

```
>plot2d("3*x^2-1","3*x^3-1",xmin=-1/sqrt(3),xmax=1/sqrt(3),square=1):
>sol &= radcan(ds(3*x^2-1,3*x^3-1)); $sol
>$showev('integrate(sol,x,0,1/sqrt(3))), $2*float(%) // panjang kurva di atas
```

LATIHAN: KOORDINAT KARTESIUS

Soal 1

Fungsi: $y = x^2 - x + 1$, $x \in [0,2]$

```
>function f1(x) &= x^2 - x + 1; $'y=f1(x)
>plot2d("x^2 - x + 1",0,2):
```

Soal 2 Fungsi: $y = \sin(2x)$, $x \in [0, \pi]$

```
>function f2(x) &= sin(2*x); $'y=f2(x)
>plot2d("sin(2*x)",0,pi):
```

Soal 3

Fungsi: $y = \ln(x+1)$, $x \in [0, 3]$

```
>function f3(x) &= ln(x+1); $'y=f3(x)
>plot2d("ln(x+1)",0,3):
```

Soal 4

Fungsi: $y = \sqrt{4-x^2}$, $x \in [0, 2]$

```
>function f4(x) &= sqrt(4-x^2); $'y=f4(x)
>plot2d("sqrt(4-x^2)",0,2):
```

Soal 5

Fungsi: $y = e^{-x} \cos(x)$, $x \in [0, 2\pi]$

```
>function f5(x) &= exp(-x)*cos(x); $'y=f5(x)
>plot2d("exp(-x)*cos(x)",0,2*pi):
```

Sikloid

Berikut kita akan menghitung panjang kurva lintasan (sikloid) suatu titik pada lingkaran yang berputar ke kanan pada permukaan datar. Misalkan jari-jari lingkaran tersebut adalah r . Posisi titik pusat lingkaran pada saat t adalah:

$$(rt, r).$$

Misalkan posisi titik pada lingkaran tersebut mula-mula $(0,0)$ dan posisinya pada saat t adalah:

$$(r(t - \sin(t)), r(1 - \cos(t))).$$

Berikut kita plot lintasan tersebut dan beberapa posisi lingkaran ketika $t=0$, $t=\pi/2$, $t=\pi$.

```
>x &= r*(t-sin(t))
```

$r(t - \sin(t))$

```
>y &= r*(1-cos(t))
```

$r(1 - \cos(t))$

Berikut kita gambar sikloid untuk $r=1$.

```
>ex &= x-sin(x); ey &= 1-cos(x); aspect(1);  
>plot2d(ex,ey,xmin=0,xmax=4pi,square=1); ...  
> plot2d("2+cos(x)","1+sin(x)",xmin=0,xmax=2pi,>add,color=blue); ...  
> plot2d([2,ex(2)], [1,ey(2)],color=red,>add); ...  
> plot2d(ex(2),ey(2),>points,>add,color=red); ...  
> plot2d("2pi+cos(x)","1+sin(x)",xmin=0,xmax=2pi,>add,color=blue); ...  
> plot2d([2pi,ex(2pi)], [1,ey(2pi)],color=red,>add); ...  
> plot2d(ex(2pi),ey(2pi),>points,>add,color=red):
```

Berikut dihitung panjang lintasan untuk 1 putaran penuh. (Jangan salah menduga bahwa panjang lintasan 1 putaran penuh sama dengan keliling lingkaran!)

```
>ds &= radcan(sqrt(diff(ex,x)^2+diff(ey,x)^2)); $ds=trigsimp(ds) // elemen panjang kurva sikloid
>ds &= trigsimp(ds); $ds
>$showev('integrate(ds,x,0,2*pi)) // hitung panjang sikloid satu putaran penuh
>integrate(mxm("ds"),0,2*pi) // hitung secara numerik
```

8

```
>romberg(mxm("ds"),0,2*pi) // cara lain hitung secara numerik
```

8

Perhatikan, seperti terlihat pada gambar, panjang sikloid lebih besar daripada keliling lingkarannya, yakni:

$$2\pi.$$

Kurvatur (Kelengkungan) Kurva

image: Osculating.png

Aslinya, kelengkungan kurva diferensiabel (yakni, kurva mulus yang tidak lancip) di titik P didefinisikan melalui lingkaran oskulasi (yaitu, lingkaran yang melalui titik P dan terbaik memperkirakan, paling banyak menyinggung kurva di sekitar P). Pusat dan radius kelengkungan kurva di P adalah pusat dan radius lingkaran oskulasi. Kelengkungan adalah kebalikan dari radius kelengkungan:

$$\kappa = \frac{1}{R}$$

dengan R adalah radius kelengkungan. (Setiap lingkaran memiliki kelengkungan ini pada setiap titiknya, dapat diartikan, setiap lingkaran berputar 2π sejauh $2\pi R$.)

Definisi ini sulit dimanipulasi dan dinyatakan ke dalam rumus untuk kurva umum. Oleh karena itu digunakan definisi lain yang ekuivalen.

Definisi Kurvatur dengan Fungsi Parametrik Panjang Kurva

Setiap kurva diferensiabel dapat dinyatakan dengan persamaan parametrik terhadap panjang kurva s:

$$\gamma(s) = (x(s), y(s)),$$

dengan x dan y adalah fungsi riil yang diferensiabel, yang memenuhi:

$$\|\gamma'(s)\| = \sqrt{x'(s)^2 + y'(s)^2} = 1.$$

Ini berarti bahwa vektor singgung

$$\mathbf{T}(s) = (x'(s), y'(s))$$

memiliki norm 1 dan merupakan vektor singgung satuan.

Apabila kurvanya memiliki turunan kedua, artinya turunan kedua x dan y ada, maka $\mathbf{T}'(s)$ ada. Vektor ini merupakan normal kurva yang arahnya menuju pusat kurvatur, norm-nya merupakan nilai kurvatur (kelengkungan):

$$\begin{aligned}\mathbf{T}(s) &= \gamma'(s), \\ \mathbf{T}^2(s) &= 1 \text{ (konstanta)} \Rightarrow \mathbf{T}'(s) \cdot \mathbf{T}(s) = 0 \\ \kappa(s) &= \|\mathbf{T}'(s)\| = \|\gamma''(s)\| = \sqrt{x''(s)^2 + y''(s)^2}.\end{aligned}$$

Nilai

$$R(s) = \frac{1}{\kappa(s)}$$

disebut jari-jari (radius) kelengkungan kurva.

Bilangan riil

$$k(s) = \pm \kappa(s)$$

disebut nilai kelengkungan bertanda.

Contoh:

Akan ditentukan kurvatur lingkaran

$$x = r \cos t, \quad y = r \sin t.$$

```
>fx &= r*cos(t); fy &=r*sin(t);  
>&assume(t>0,r>0); s &=integrate(sqrt(diff(fx,t)^2+diff(fy,t)^2),t,0,t); s // elemen panjang kurva,
```

$$r \, t$$

```
>&kill(s); fx &= r*cos(s/r); fy &=r*sin(s/r); // definisi ulang persamaan parametrik terhadap s deng  
>k &= trigsimp(sqrt(diff(fx,s,2)^2+diff(fy,s,2)^2)); $k // nilai kurvatur lingkaran dengan menggunak
```

Untuk representasi parametrik umum, misalkan

$$x = x(t), \quad y = y(t)$$

merupakan persamaan parametrik untuk kurva bidang yang terdiferensialkan dua kali. Kurvatur untuk kurva tersebut didefinisikan sebagai

$$\kappa = \frac{d\phi}{ds} = \frac{\frac{d\phi}{dt}}{\frac{ds}{dt}} \quad (\phi \text{ adalah sudut kemiringan garis singgung dan } s \text{ adalah panjang kurva})$$
$$= \frac{\frac{d\phi}{dt}}{\sqrt{\left(\frac{dx}{dt}\right)^2 + \left(\frac{dy}{dt}\right)^2}} = \frac{\frac{d\phi}{dt}}{\sqrt{x'(t)^2 + y'(t)^2}}.$$

Selanjutnya, pembilang pada persamaan di atas dapat dicari sebagai berikut.

$$\sec^2 \phi \frac{d\phi}{dt} = \frac{d}{dt} (\tan \phi) = \frac{d}{dt} \left(\frac{dy}{dx} \right) = \frac{d}{dt} \left(\frac{dy/dt}{dx/dt} \right) = \frac{d}{dt} \left(\frac{y'(t)}{x'(t)} \right) = \frac{x'(t)y''(t) - x''(t)y'(t)}{x'(t)^2}.$$

$$\begin{aligned} \frac{d\phi}{dt} &= \frac{1}{\sec^2 \phi} \frac{x'(t)y''(t) - x''(t)y'(t)}{x'(t)^2} \\ &= \frac{1}{1 + \tan^2 \phi} \frac{x'(t)y''(t) - x''(t)y'(t)}{x'(t)^2} \\ &= \frac{1}{1 + \left(\frac{y'(t)}{x'(t)} \right)^2} \frac{x'(t)y''(t) - x''(t)y'(t)}{x'(t)^2} \\ &= \frac{x'(t)y''(t) - x''(t)y'(t)}{x'(t)^2 + y'(t)^2}. \end{aligned}$$

Jadi, rumus kurvatur untuk kurva parametrik

$$x = x(t), \quad y = y(t)$$

adalah

$$\kappa(t) = \frac{x'(t)y''(t) - x''(t)y'(t)}{(x'(t)^2 + y'(t)^2)^{3/2}}.$$

Jika kurvanya dinyatakan dengan persamaan parametrik pada koordinat kutub

$$x = r(\theta) \cos \theta, \quad y = r(\theta) \sin \theta,$$

maka rumus kurvturnya adalah

$$\kappa(\theta) = \frac{r(\theta)^2 + 2r'(\theta)^2 - r(\theta)r''(\theta)}{(r'(\theta)^2 + r(\theta)^2)^{3/2}}.$$

(Silakan Anda turunkan rumus tersebut!)

Contoh:

Lingkar dengan pusat (0,0) dan jari-jari r dapat dinyatakan dengan persamaan parametrik

$$x = r \cos t, \quad y = r \sin t.$$

Nilai kelengkungan lingkaran tersebut adalah

$$\kappa(t) = \frac{x'(t)y''(t) - x''(t)y'(t)}{(x'(t)^2 + y'(t)^2)^{3/2}} = \frac{r^2}{r^3} = \frac{1}{r}.$$

Hasil cocok dengan definisi kurvatur suatu kelengkungan.

Kurva

$$y = f(x)$$

dapat dinyatakan ke dalam persamaan parametrik

$$x = t, \quad y = f(t), \quad \text{dengan } x'(t) = 1, \quad x''(t) = 0,$$

sehingga kurvturnya adalah

$$\kappa(t) = \frac{y''(t)}{(1 + y'(t)^2)^{3/2}}.$$

Contoh:

Akan ditentukan kurvatur parabola

$$y = ax^2 + bx + c.$$

```
>function f(x) &= a*x^2+b*x+c; $y=f(x)
>function k(x) &= (diff(f(x),x,2))/(1+diff(f(x),x)^2)^(3/2); $'k(x)=k(x) // kelengkungan parabola
>function f(x) &= x^2+x+1; $y=f(x) // akan kita plot kelengkungan parabola untuk a=b=c=1
>function k(x) &= (diff(f(x),x,2))/(1+diff(f(x),x)^2)^(3/2); $'k(x)=k(x) // kelengkungan parabola
```

Berikut kita gambar parabola tersebut beserta kurva kelengkungan, kurva jari-jari kelengkungan dan salah satu lingkaran oskulasi di titik puncak parabola. Perhatikan, puncak parabola dan jari-jari lingkaran oskulasi di puncak parabola adalah

$$(-1/2, 3/4), 1/k(2) = 1/2,$$

sehingga pusat lingkaran oskulasi adalah $(-1/2, 5/4)$.

```
>plot2d(["f(x)", "k(x)"],-2,1, color=[blue,red]); plot2d("1/k(x)",-1.5,1,color=green,>add); ...
>plot2d("-1/2+1/k(-1/2)*cos(x)", "5/4+1/k(-1/2)*sin(x)",xmin=0,xmax=2pi,>add,color=blue):
```

Untuk kurva yang dinyatakan dengan fungsi implisit

$$F(x, y) = 0$$

dengan turunan-turunan parsial

$$F_x = \frac{\partial F}{\partial x}, \quad F_y = \frac{\partial F}{\partial y}, \quad F_{xy} = \frac{\partial}{\partial y} \left(\frac{\partial F}{\partial x} \right), \quad F_{xx} = \frac{\partial}{\partial x} \left(\frac{\partial F}{\partial x} \right), \quad F_{yy} = \frac{\partial}{\partial y} \left(\frac{\partial F}{\partial y} \right),$$

berlaku

$$F_x dx + F_y dy = 0 \text{ atau } \frac{dy}{dx} = -\frac{F_x}{F_y},$$

sehingga kurvturnya adalah

$$\kappa = \frac{F_y^2 F_{xx} - 2F_x F_y F_{xy} + F_x^2 F_{yy}}{(F_x^2 + F_y^2)^{3/2}}.$$

(Silakan Anda turunkan sendiri!)

Contoh 1:

Parabola

$$y = ax^2 + bx + c$$

dapat dinyatakan ke dalam persamaan implisit

$$ax^2 + bx + c - y = 0.$$

```
>function F(x,y) &=a*x^2+b*x+c-y; $F(x,y)
>Fx &= diff(F(x,y),x), Fxx &=diff(F(x,y),x,2), Fy &=diff(F(x,y),y), Fxy &=diff(diff(F(x,y),x),y), Fy
```

$$2ax + b$$

$$2a$$

$$-1$$

$$0$$

$$0$$

```
>function k(x) &= (Fy^2*Fxx-2*Fx*Fy*Fxy+Fx^2*Fyy)/(Fx^2+Fy^2)^(3/2); $'k(x)=k(x) // kurvatur parabol
```

Hasilnya sama dengan sebelumnya yang menggunakan persamaan parabola biasa.

LATIHAN INTEGRAL DAN KURVA

1. Integral tentu dan tak tentu Pilih minimal 5 fungsi berbeda dari buku Kalkulus, misal:

```
>function f1(x) &= x^2 + 3*x + 2; $'f1(x)=f1(x)
>function f2(x) &= sin(x); $'f2(x)=f2(x)
>function f3(x) &= exp(-x); $'f3(x)=f3(x)
>function f4(x) &= 1/(1+x^2); $'f4(x)=f4(x)
>function f5(x) &= sqrt(1-x^2); $'f5(x)=f5(x)
```

Hitung anti turunan (integral tak tentu)

```
>$showev('integrate(f1(x),x))
>$showev('integrate(f2(x),x))
>$showev('integrate(f3(x),x))
>$showev('integrate(f4(x),x))
>$showev('integrate(f5(x),x))
```

Integral tentu dengan batas menarik

```
>$showev('integrate(f1(x),x,0,2))  
>$showev('integrate(f2(x),x,0,pi))  
>$showev('integrate(f3(x),x,0,1))  
>$showev('integrate(f4(x),x,0,1))  
>$showev('integrate(f5(x),x,0,1))
```

Gambar grafik fungsi dan daerah integrasi

```
>plot2d("x^2+3*x+2",0,2)  
>plot2d("sin(x)",0,pi,>add)  
>plot2d("exp(-x)",0,1,>add):  
>plot2d("1/(1+x^2)",0,1,>add)  
>plot2d("sqrt(1-x^2)",0,1,>add):
```

Barisan dan Deret

Barisan adalah susunan bilangan yang mengikuti pola tertentu. Contoh: 2, 4, 6, 8, 10, ... adalah barisan aritmetika dengan beda 2. Sedangkan deret adalah jumlah dari suku-suku barisan tersebut. Misalnya: $2 + 4 + 6 + 8 + 10 = 30$.

Dalam EMT (Euler Math Toolbox), barisan dapat didefinisikan dengan berbagai cara:

- menggunakan tanda titik dua (:) untuk membentuk vektor beraturan - menggunakan "sequence" atau "makelist" dari Maxima - menggunakan fungsi-fungsi seperti "sum", "cumsum", dan "differences"

Contoh penggunaan dasar EMT untuk barisan dan deret:

```
> 1:10 // barisan sederhana dari 1 sampai 10
```

```
[1, 2, 3, 4, 5, 6, 7, 8, 9, 10]
```

```
> 1:2:30 // barisan dengan beda 2
```

```
[1, 3, 5, 7, 9, 11, 13, 15, 17, 19, 21, 23, 25, 27, 29]
```

CONTOH SOAL 1: Barisan Aritmetika

Diketahui barisan aritmetika: 3, 6, 9, 12, ... Tentukan suku ke-10 dan jumlah 10 suku pertama.

Rumus suku ke- n : $a_n = a_1 + (n - 1)b$ Rumus jumlah n suku: $S_n = n/2 * (2a_1 + (n - 1)b)$ Diketahui: $a_1 = 3$, $b = 3$, $n = 10$

```
> a1 = 3; b = 3; n = 10;  
> an = a1 + (n - 1)*b; an // suku ke-10
```

30

```
> S10 = n/2 * (2*a1 + (n - 1)*b); S10 // jumlah 10 suku pertama
```

165

Jadi, suku ke-10 adalah 30 dan jumlah 10 suku pertama adalah 165.

CONTOH SOAL 2: Barisan Geometri

Diketahui barisan geometri: 2, 6, 18, 54, ... Tentukan suku ke-6 dan jumlah 6 suku pertama.

Rumus suku ke- n : $a_n = a_1 * r^{(n-1)}$ Rumus jumlah n suku: $S_n = a_1 * (r^n - 1)/(r - 1)$ Diketahui: $a_1 = 2$, $r = 3$, $n = 6$

```
> a1 = 2; r = 3; n = 6;  
> an = a1 * r^(n - 1); an // suku ke-6
```

486

```
> S6 = a1 * (r^n - 1)/(r - 1); S6 // jumlah 6 suku pertama
```

728

Jadi, suku ke-6 adalah 486 dan jumlah 6 suku pertama adalah 728.

CONTOH SOAL 3: Menggunakan Fungsi EMT “sum” dan “differences”

Misalkan barisan bilangan genap dari 2 hingga 20:

```
> v = 2:2:20; v
```

```
[2, 4, 6, 8, 10, 12, 14, 16, 18, 20]
```

Hitung jumlah semua elemen dan selisih antar elemen berturutan.

```
> sum(v)
```

```
110
```

```
> differences(v)
```

```
[2, 2, 2, 2, 2, 2, 2, 2, 2]
```

Jadi, jumlah semua elemen barisan adalah 110 dan beda antar elemen adalah 2.

CONTOH SOAL 4: Deret Tak Hingga

Deret geometri tak hingga dengan $a_1 = 5$ dan $r = 1/2$. Rumus jumlah tak hingga: $S = a_1 / (1 - r)$

```
> a1 = 5; r = 1/2;  
> S = a1 / (1 - r); S
```

Iterasi dan Barisan

Iterasi adalah proses menghasilkan nilai-nilai berurutan menggunakan suatu aturan (fungsi) yang berulang.

Dalam EMT, digunakan fungsi:

```
iterate("g(x)", x0, n)
```

untuk menghasilkan barisan:

```
x_(k+1) = g(x_k), dengan x_0 nilai awal.
```

Contohnya digunakan dalam masalah bunga, pertumbuhan populasi, tabungan, dsb.

CONTOH SOAL 1: Pertumbuhan Nilai

Seorang investor menanam modal Rp1.000 dengan kenaikan 5% tiap periode. Hitung nilai investasinya setelah 10 periode.

```
> q = 1.05;  
> iterate("x*q", 1000, n=10)'
```

```
1000  
1050  
1102.5  
1157.63  
1215.51  
1276.28  
1340.1  
1407.1  
1477.46  
1551.33  
1628.89
```

Penjelasan: Iterasi menghasilkan barisan nilai modal setiap periode. Nilai akhir setelah 10 periode adalah Rp 1.628,89.

CONTOH SOAL 2: Tabungan dan Biaya Administrasi

Misal bunga tabungan 0,5% per bulan, pajak bunga 20%, dan biaya admin Rp10.000/bulan. Tabungan awal Rp1.000.000. Berapa sisa saldo setelah 130 bulan?

```
> r = 0.005;  
> plot2d(iterate("(1+0.8*r)*x-10000",1000000,n=130)):
```

Penjelasan: $-(1+0.8*r)*x$: bunga dikalikan 80% (setelah pajak) - 10000 : biaya admin bulanan Grafik akan menunjukkan saldo yang makin lama makin berkurang. Anda bisa ubah nilai awal tabungan untuk melihat efeknya.

CONTOH SOAL 3: Menentukan Tabungan Minimum Agar Aman

Hitung saldo minimum agar tidak berkurang dengan bunga 0.5% dan biaya admin 10 ribu.

Rumusnya diperoleh dari:

$$0.8*r*A - a = 0$$

Sehingga:

$$A = a / (0.8*r)$$

```
> $solve(0.8*r*A - a, A), $% with [r=0.005, a=10]
```

Dari hasil Maxima, tabungan minimal sekitar 2,5 juta rupiah.

CONTOH SOAL 4: Fungsi Saldo dan Iterasi Ulang

Kita definisikan fungsi saldo:

```
> function saldo(x, r, a) := round((1 + 0.8*r)*x - a, 2);
```

Lalu kita uji dengan tabungan awal berbeda.

```
> iterate({"saldo", 0.005, 10}}, 1000, n=6)
```

```
[1000, 994, 987.98, 981.93, 975.86, 969.76, 963.64]
```

```
> iterate({"saldo", 0.005, 10}}, 2000, n=6)
```

```
[2000, 1998, 1995.99, 1993.97, 1991.95, 1989.92, 1987.88]
```

Spiral Theodorus

Spiral Theodorus (spiral segitiga siku-siku) dapat digambar secara rekursif. Spiral ini pertama kali diperkenalkan oleh Theodorus dari Cyrene, seorang matematikawan Yunani kuno. Setiap segitiga baru dibentuk dengan sisi miring dari segitiga sebelumnya sebagai salah satu sisinya.

Rumus rekursif spiral Theodorus dalam bentuk bilangan kompleks adalah:

$$x_n = \left(1 + \frac{i}{\sqrt{n-1}}\right) x_{n-1}, \quad x_1 = 1$$

Dengan i adalah bilangan imajiner, dan setiap hasil perkalian kompleks akan menghasilkan titik baru yang berputar dengan jarak tetap pada bidang kompleks.

Contoh Soal 1: Buatlah 17 titik Spiral Theodorus pertama dan tampilkan hasil plotnya.

Langkah pengerjaan: 1. Definisikan fungsi kompleks rekursif $g(n)$ 2. Gunakan perintah "sequence" untuk menghasilkan barisan kompleks. 3. Gambar spiral menggunakan "plot2d".

```
>function g(n) := 1+I/sqrt(n)
>x=sequence("g(n-1)*x[n-1]",1,17)
```

```
[ 1+0i , 1+1i, 0.292893+1.70711i, -0.692705+1.87621i,
-1.63081+1.52986i, -2.31498+0.800536i, -2.6418-0.144552i,
-2.58716-1.14306i, -2.18303-2.05776i, -1.49711-2.78544i,
-0.61628-3.25886i, 0.366304-3.44468i, 1.3607-3.33894i,
2.28675-2.96155i, 3.07826-2.35039i, 3.68513-1.55558i,
4.07402-0.634302i ]
```

```
>plot2d(x,r=3.5):
>textbox(latex("Spiral\ Theodorus"),0.4)
```

Contoh Soal 2: Hubungkan titik pusat (0,0) dengan setiap titik hasil spiral agar membentuk garis radial. Gunakan loop untuk menggambar tiap segmen garis.

```
>for i=1:cols(x); plot2d([0,x[i]],>add); end:
```

Contoh Soal 3: Spiral Theodorus juga bisa didefinisikan menggunakan fungsi vektor (tanpa bilangan kompleks). Rumusnya: setiap langkah baru adalah hasil rotasi vektor sebelumnya sejauh sudut tertentu. Berikut fungsi gstep yang melakukan rotasi dan normalisasi panjang vektor:

```
>function gstep (v)

w=[-v[2];v[1]]
return v+w/norm(w)
endfunction
```

Sekarang lakukan iterasi sebanyak 16 kali dimulai dari vektor $[1;0]$.

```
>x=iterate("gstep",[1;0],16)
```

```
0
1
-1
1
-1.70711
0.292893
-1.87621
-0.692705
-1.52986
-1.63081
-0.800536
-2.31498
0.144552
-2.6418
1.14306
-2.58716
2.05776
-2.18303
2.78544
-1.49711
3.25886
-0.61628
3.44468
0.366304
3.33894
1.3607
2.96155
2.28675
2.35039
```

```
3.07826
1.55558
3.68513
Real 2 x 17 matrix
      1      1      0.292893      -0.692705      ...
      0      1      1.70711      1.87621      ...
```

```
>plot2d(x[1],x[2],r=3.5,>points):
```

Contoh Soal 4: Hitung panjang sisi miring ke-10 spiral Theodorus. Diketahui bahwa panjang sisi miring ke-n adalah $\sqrt{n}$.

```
>sqrt(10):
```

Kekonvergenan

Terkadang kita ingin melakukan iterasi sampai konvergen. Apabila iterasinya tidak konvergen setelah ditunggu lama, Anda dapat menghentikannya dengan menekan tombol [ESC].

```
>iterate("cos(x)",1) // iterasi  $x(n+1)=\cos(x(n))$ , dengan  $x(0)=1$ .
```

```
0.739085133216
```

Iterasi tersebut konvergen ke penyelesaian persamaan

$$x = \cos(x).$$

Iterasi ini juga dapat dilakukan pada interval, hasilnya adalah barisan interval yang memuat akar tersebut.

```
>hasil := iterate("cos(x)",~1,2~) //iterasi  $x(n+1)=\cos(x(n))$ , dengan interval awal (1, 2
```

```
~0.739085133211,0.7390851332133~
```

Jika interval hasil tersebut sedikit diperlebar, akan terlihat bahwa interval tersebut memuat akar persamaan $x=\cos(x)$.

```
>h=expand(hasil,100), cos(h) << h
```

```
~0.73908513309,0.7390851333~  
1
```

Iterasi juga dapat digunakan pada fungsi yang didefinisikan.

```
>function f(x) := (x+2/x)/2
```

Iterasi $x_{(n+1)}=f(x_{(n)})$ akan konvergen ke akar kuadrat 2.

```
>iterate("f",2), sqrt(2)
```

```
1.41421356237  
1.41421356237
```

Jika pada perintah `iterate` diberikan tambahan parameter `n`, maka hasil iterasinya akan ditampilkan mulai dari iterasi pertama sampai ke-`n`.

```
>iterate("f",2,5)
```

```
[2, 1.5, 1.41667, 1.41422, 1.41421, 1.41421]
```

Untuk iterasi ini tidak dapat dilakukan terhadap interval.

```
>niterate("f",~1,2~,5)
```

```
[ ~1,2~, ~1,2~, ~1,2~, ~1,2~, ~1,2~, ~1,2~ ]
```

Perhatikan, hasil iterasinya sama dengan interval awal. Alasannya adalah perhitungan dengan interval bersifat terlalu longgar. Untuk meningkatkan perhitungan pada ekspresi dapat digunakan pembagian intervalnya, menggunakan fungsi `ieval()`.

```
>function s(x) := ieval("(x+2/x)/2",x,10)
```

Selanjutnya dapat dilakukan iterasi hingga diperoleh hasil optimal, dan intervalnya tidak semakin mengecil. Hasilnya berupa interval yang memuat akar persamaan:

$$x = \frac{1}{2} \left(x + \frac{2}{x} \right).$$

Satu-satunya solusi adalah

$$x = \sqrt{2}.$$

```
>iterate("s",~1,2~)
```

```
~1.41421356236,1.41421356239~
```

Fungsi "iterate()" juga dapat bekerja pada vektor. Berikut adalah contoh fungsi vektor, yang menghasilkan rata-rata aritmetika dan rata-rata geometri.

$$(a_{n+1}, b_{n+1}) = \left(\frac{a_n + b_n}{2}, \sqrt{a_n b_n} \right)$$

Iterasi ke-n disimpan pada vektor kolom x[n].

```
>function g(x) := [(x[1]+x[2])/2;sqrt(x[1]*x[2])]
```

Iterasi dengan menggunakan fungsi tersebut akan konvergen ke rata-rata aritmetika dan geometri dari nilai-nilai awal.

```
>iterate("g",[1;5])
```

```
2.60401
2.60401
```

Hasil tersebut konvergen agak cepat, seperti kita cek sebagai berikut.

```
>iterate("g",[1;5],4)
```

1	3	2.61803	2.60403	2.60401
5	2.23607	2.59002	2.60399	2.60401

Iterasi pada interval dapat dilakukan dan stabil, namun tidak menunjukkan bahwa limitnya pada batas-batas yang dihitung.

```
>iterate("g",[~1~;~5~],4)
```

Interval 2 x 5 matrix

```
~0.999999999999999778,1.000000000000000022~ ...
~4.99999999999999911,5.000000000000000089~ ...
```

Iterasi berikut konvergen sangat lambat.

$$x_{n+1} = \sqrt{x_n}.$$

```
>iterate("sqrt(x)",2,10)
```

```
[2, 1.41421, 1.18921, 1.09051, 1.04427, 1.0219, 1.01089,  
1.00543, 1.00271, 1.00135, 1.00068]
```

Kekonvergenan iterasi tersebut dapat dipercepat dengan percepatan Steffenson:

```
>steffenson("sqrt(x)",2,10)
```

```
[1.04888, 1.00028, 1, 1]
```

Iterasi menggunakan Loop yang ditulis Langsung

Berikut adalah beberapa contoh penggunaan loop untuk melakukan iterasi yang ditulis langsung pada baris perintah.

```
>x=2; repeat x=(x+2/x)/2; until x^2~=2; end; x,
```

```
1.41421356237
```

Penggabungan matriks menggunakan tanda "|" dapat digunakan untuk menyimpan semua hasil iterasi.

```
>v=[1]; for i=2 to 8; v=v|(v[i-1]*i); end; v,
```

```
[1, 2, 6, 24, 120, 720, 5040, 40320]
```

hasil iterasi juga dapat disimpan pada vektor yang sudah ada.

```
>v=ones(1,100); for i=2 to cols(v); v[i]=v[i-1]*i; end; ...  
>plot2d(v,logplot=1); textbox(latex(&log(n)),x=0.5):  
>A =[0.5,0.2;0.7,0.1]; b=[2;2]; ...  
>x=[1;1]; repeat xnew=A.x-b; until all(xnew~=x); x=xnew; end;...  
>x,
```

-7.09677

-7.74194

Iterasi di dalam Fungsi

Fungsi atau program juga dapat menggunakan iterasi dan dapat digunakan untuk melakukan iterasi. Berikut adalah beberapa contoh iterasi di dalam fungsi.

Contoh berikut adalah suatu fungsi untuk menghitung berapa lama suatu iterasi konvergen. Nilai fungsi tersebut adalah hasil akhir iterasi dan banyak iterasi sampai konvergen.

```
>function map hiter(f$,x0) ...
```

```
  x=x0;
  maxiter=0;
  repeat
    xnew=f$(x);
    maxiter=maxiter+1;
    until xnew~=x;
    x=xnew;
  end;
  return maxiter;
endfunction
```

Misalnya, berikut adalah iterasi untuk mendapatkan hampiran akar kuadrat 2, cukup cepat, konvergen pada iterasi ke-5, jika dimulai dari hampiran awal 2.

```
>hiter("(x+2/x)/2",2)
```

Karena fungsinya didefinisikan menggunakan "map". maka nilai awalnya dapat berupa vektor.

```
>x=1.5:0.1:10; hasil=hiter("(x+2/x)/2",x); ...  
> plot2d(x,hasil):
```

Dari gambar di atas terlihat bahwa kekonvergenan iterasinya semakin lambat, untuk nilai awal semakin besar, namun penambahannya tidak kontinu. Kita dapat menemukan kapan maksimum iterasinya bertambah.

```
>hasil[1:10]
```

```
[4, 5, 5, 5, 5, 5, 6, 6, 6, 6]
```

```
>x[nonzeros(differences(hasil))]
```

```
[1.5, 2, 3.4, 6.6]
```

maksimum iterasi sampai konvergen meningkat pada saat nilai awalnya 1.5, 2, 3.4, dan 6.6.

Contoh berikutnya adalah metode Newton pada polinomial kompleks berderajat 3.

```
>p = x^3-1; newton = x-p/diff(p,x); $newton
```

Selanjutnya didefinisikan fungsi untuk melakukan iterasi (aslinya 10 kali).

```
>function iterasi(f$,x,n=10) ...
```

```
    loop 1 to n; x=f$(x); end;  
    return x;  
endfunction
```

Kita mulai dengan menentukan titik-titik grid pada bidang kompleksnya.

```
>r=1.5; x=linspace(-r,r,501); Z=x+I*x'; W=iterasi(newton,Z);
```

Berikut adalah akar-akar polinomial di atas.

```
>z=&solve(p)()
```

```
[ -0.5+0.866025i,  -0.5-0.866025i,  1+0i  ]
```

Untuk menggambar hasil iterasinya, dihitung jarak dari hasil iterasi ke-10 ke masing-masing akar, kemudian digunakan untuk menghitung warna yang akan digambar, yang menunjukkan limit untuk masing-masing nilai awal.

Fungsi `plotrgb()` menggunakan jendela gambar terkini untuk menggambar warna RGB sebagai matriks.

```
>C=rgb(max(abs(W-z[1]),1),max(abs(W-z[2]),1),max(abs(W-z[3]),1)); ...  
> plot2d(none,-r,r,-r,r); plotrgb(C):
```

Iterasi Simbolik

Seperti sudah dibahas sebelumnya, untuk menghasilkan barisan ekspresi simbolik dengan Maxima dapat digunakan fungsi `makelist()`.

```
>&powerdisp:true // untuk menampilkan deret pangkat mulai dari suku berpangkat terkecil
```

```
true
```

```
>deret &= makelist(taylor(exp(x),x,0,k),k,1,3); $deret // barisan deret Taylor untuk  $e^x$ 
```

Untuk mengubah barisan deret tersebut menjadi vektor string di EMT digunakan fungsi `mxm2str()`. Selanjutnya, vektor string/ekspresi hasilnya dapat digambar seperti menggambar vektor ekspresi pada EMT.

```
>plot2d("exp(x)",0,3); // plot fungsi aslinya,  $e^x$   
>plot2d(mxm2str("deret"),>add,color=4:6): // plot ketiga deret taylor hampiran fungsi tersebut
```

Selain cara di atas dapat juga dengan cara menggunakan indeks pada vektor/list yang dihasilkan.

```
>$deret[3]  
>plot2d(["exp(x)",&deret[1],&deret[2],&deret[3]],0,3,color=1:4):  
>$sum(sin(k*x)/k,k,1,5)
```

Berikut adalah cara menggambar kurva

$$y = \sin(x) + \frac{\sin 3x}{3} + \frac{\sin 5x}{5} + \dots$$

```
>plot2d(&sum(sin((2*k+1)*x)/(2*k+1),k,0,20),0,2pi):
```

Hal serupa juga dapat dilakukan dengan menggunakan matriks, misalkan kita akan menggambar kurva

$$y = \sum_{k=1}^{100} \frac{\sin(kx)}{k}, \quad 0 \leq x \leq 2\pi.$$

```
>x=linspace(0,2pi,1000); k=1:100; y=sum(sin(k*x')/k)'; plot2d(x,y):
```

Tabel Fungsi

Terdapat cara menarik untuk menghasilkan barisan dengan ekspresi Maxima. Perintah `mxmtable()` berguna untuk menampilkan dan menggambar barisan dan menghasilkan barisan sebagai vektor kolom.

Sebagai contoh berikut adalah barisan turunan ke- n x^x di $x=1$.

```
>mxmtable("diffat(x^x,x=1,n)","n",1,8,frac=1);
```

```
1
2
3
8
10
54
-42
944
```

```
>$'sum(k, k, 1, n) = factor(ev(sum(k, k, 1, n),simpsum=true)) // simpsum:menghitung deret secara sim
>$'sum(1/(3^k+k), k, 0, inf) = factor(ev(sum(1/(3^k+k), k, 0, inf),simpsum=true))
```

Di sini masih gagal, hasilnya tidak dihitung.

```
>$'sum(1/x^2, x, 1, inf)= ev(sum(1/x^2, x, 1, inf),simpsum=true) // ev: menghitung nilai ekspresi
>$'sum((-1)^(k-1)/k, k, 1, inf) = factor(ev(sum((-1)^(x-1)/x, x, 1, inf),simpsum=true))
```

Di sini masih gagal, hasilnya tidak dihitung.

```
>$'sum((-1)^k/(2*k-1), k, 1, inf) = factor(ev(sum((-1)^k/(2*k-1), k, 1, inf),simpsum=true))
>$ev(sum(1/n!, n, 0, inf),simpsum=true)
```

Di sini masih gagal, hasilnya tidak dihitung, harusnya hasilnya e.

```
>&assume(abs(x)<1); $'sum(a*x^k, k, 0, inf)=ev(sum(a*x^k, k, 0, inf),simpsum=true), &forget(abs(x)<1)
```

Deret geometri tak hingga, dengan asumsi rasional antara -1 dan 1.

```
>$'sum(x^k/k!,k,0,inf)=ev(sum(x^k/k!,k,0,inf),simpsum=true)
>$limit(sum(x^k/k!,k,0,n),n,inf)
>function d(n) &= sum(1/(k^2-k),k,2,n); $'d(n)=d(n)
>$d(10)=ev(d(10),simpsum=true)
>$d(100)=ev(d(100),simpsum=true)
>
```

Deret Taylor

Deret Taylor suatu fungsi f yang diferensiabel sampai tak hingga di sekitar $x=a$ adalah:

$$f(x) = \sum_{k=0}^{\infty} \frac{(x-a)^k f^{(k)}(a)}{k!}.$$

```
>$'e^x=taylor(exp(x),x,0,10) // deret Taylor e^x di sekitar x=0, sampai suku ke-11  
>$'log(x)=taylor(log(x),x,1,10)// deret log(x) di sekitar x=1  
>
```